

Excel supplement: Chapter 7

Matrix and vector algebra

Many models in economics lead to large systems of linear equations. These problems are particularly suited for computers. The main purpose of this chapter is to show you how to work with matrices and vectors in Excel, and use matrices and vectors to solve linear systems of equations.

Matrices and matrix operations

This section shows how matrices can be represented in Excel. A matrix may be thought of as a two-dimensional table. As such, matrices can easily be represented in a spreadsheet.

	A	B	C	D	E	F	G	H
1		A					C	
2			3	-2			-1	2
3			5	8			8	5
4							7	6
5							1	1
6		B						
7			-1	2	1,732051	16		

Figure 1. Defining matrices in Excel

The matrices in Figure 1 correspond with the matrices in Example 15.2.1 of the textbook. Note that the matrices are enclosed in brackets. We will adopt a convention that, whenever it may be unclear what matrix we are referring to in a figure, it is enclosed with lines on the left and right-hand sides, and it is named in the upper left corner. In the matrix **B**, $1.732051 \approx \sqrt{3}$ is entered by entering the formula =SQRT(3).

Often we must enter numbers manually, but sometimes we can use formulas to simplify the process. Let us construct a matrix whose ij -th element is $a_{ij} = 2i - j$, where i is the row number and j is the column number. (Example 15.2.2 of the textbook.) We first create two index vectors and use them to create formulas for each element.

	C	D	E	F	G
10					
11		j	1	2	3
12	i				
13	1		=2*\$C13-E\$11	=2*\$C13-F\$11	=2*\$C13-G\$11
14	2		=2*\$C14-E\$11	=2*\$C14-F\$11	=2*\$C14-G\$11
15	3		=2*\$C15-E\$11	=2*\$C15-F\$11	=2*\$C15-G\$11
16	4		=2*\$C16-E\$11	=2*\$C16-F\$11	=2*\$C16-G\$11
17					

Figure 2. Creating matrices using formulas

Using a mix of relative and absolute references¹ allows us to enter the formula only once, for example in cell E13, and copy the contents in each cell to the rest of the cells in E13:G16. The resulting matrix looks like Figure 3.

	C	D	E	F	G	H
10						
11			1	2	3	
12						
13	1		1	0	-1	
14	2		3	2	1	
15	3		5	4	3	
16	4		7	6	5	
17						

Figure 3

When working with matrices in Excel, one must use special array formulas and array functions. There are special rules that apply to these functions. The reason for this is that array functions return arrays. If we add two 3×3 matrices the result is a three by three matrix. When entering the formula for adding two matrices we must take this into account. We do this by entering the formula for adding two matrices directly into a matrix. As an example, consider the following summation:

$$\begin{bmatrix} 1 & -2 \\ 2 & 1 \end{bmatrix} + \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} = \begin{bmatrix} 2 & -2 \\ 2 & 2 \end{bmatrix}$$

In order to perform this summation we must perform the following steps.

- 1) Enter the matrices into the spreadsheet.
- 2) Select empty cells so that a 2×2 range is highlighted.
- 3) Enter a formula that adds the two ranges. Either enter `=B4:C5+E4:F5` directly or enter `=`, then select the first matrix. Then enter `+` and select the second matrix. Do *not* press `<Enter>`. At this point, the spreadsheet should look like Figure 4. Note that the entire range B8:C9 is selected.
- 4) Press `<Ctrl> + <Shift> + <Enter>`.

¹ See the section on absolute and relative references in Chapter 1 of this guide.

	A	B	C	D	E	F
1						
2						
3						
4		1	-2		1	0
5		2	1		0	1
6						
7						
8		=B4:C5+E4:F5				
9						
10						

Figure 4

If the procedure is followed correctly, the spreadsheet should now look like Figure 5.

	A	B	C	D	E	F
3						
4		1	-2		1	0
5		2	1		0	1
6						
7						
8		2	-2			
9		2	2			
10						
11						

Figure 5

Multiplying any matrix with a scalar is easily done in Excel. Let us illustrate by considering

$$4 \begin{bmatrix} 1 & -2 \\ 2 & 1 \end{bmatrix} = \begin{bmatrix} 4 & -8 \\ 8 & 4 \end{bmatrix}$$

In a manner similar to matrix addition, after entering the matrix, we must select a range that has the same dimensions as the result. In this case, we select a 2×2 range. Then we enter the array formula. Assuming that we have entered the matrix, we want to multiply 4 into the cells B4:C5. We either enter =4*B4:C5 or we enter =4* without pressing <Enter> and then select the matrix. The spreadsheet should look like Figure 6. Pressing <Ctrl> + <Shift> + <Enter> concludes the calculation.

	A	B	C	D
3				
4		1	-2	
5		2	1	
6				
7				
8		=4*B4:C5		
9				
10				

Figure 6

Problems

1) Let $\mathbf{A} = \begin{bmatrix} 0 & 1 \\ 2 & 3 \end{bmatrix}$ and $\mathbf{B} = \begin{bmatrix} 1 & -1 \\ 5 & 2 \end{bmatrix}$. Evaluate $\mathbf{A} + \mathbf{B}$, $3\mathbf{A}$ and $3\mathbf{A} + \mathbf{B}$ using Excel.

2) Let $\mathbf{A} = \begin{bmatrix} 0 & 1 & -1 \\ 2 & 3 & 7 \end{bmatrix}$ and $\mathbf{B} = \begin{bmatrix} 1 & -1 & 5 \\ 0 & 1 & 9 \end{bmatrix}$. Evaluate $\mathbf{A} + \mathbf{B}$, $\mathbf{A} - \mathbf{B}$ and $5\mathbf{A} - 3\mathbf{B}$ using Excel.

3) Verify the rules for matrix addition and multiplication by scalars using the matrices:

$$\mathbf{A} = \begin{bmatrix} 1 & 3 \\ 4 & -2 \end{bmatrix}, \mathbf{B} = \begin{bmatrix} 4 & -2 \\ -1 & 3 \end{bmatrix}, \mathbf{C} = \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix}$$

Matrix multiplication

To perform matrix multiplications in Excel, we must use the function `=mmult(arg1;arg2)`. This function takes two matrices as arguments and returns their product. To illustrate its use, let us verify the following calculation.

$$\begin{bmatrix} 1 & 3 \\ 4 & -2 \end{bmatrix} \begin{bmatrix} 4 & -2 \\ -1 & 3 \end{bmatrix} = \begin{bmatrix} 1 & 7 \\ 18 & -14 \end{bmatrix}$$

First, we enter two matrices into the spreadsheet. Then, we must select 4 cells where we want to place the answer. The 4 cells that have been selected must have dimension 2×2 . After doing this and then entering the formula, the spreadsheet will look like Figure 7.

	A	B	C	D	E	F
3						
4		1	3		4	-2
5		4	-2		-1	3
6						
7						
8		=mmult(B4:C5;E4:F5)				
9						
10						
11						

Figure 7

Pressing `<Ctrl> + <Shift> + <Enter>` completes the calculation. In order to do a proper matrix multiplication, one must use `mmult`. If we try to use `*` multiplication in the same way, we perform an element by element multiplication. To illustrate this, try to enter `=B4:C5*E4:F5` and then `<Ctrl> + <Shift> + <Enter>`. We then get the matrix:

$$\begin{bmatrix} 4 & -6 \\ -4 & -6 \end{bmatrix} = \begin{bmatrix} 1 \cdot 4 & 3 \cdot (-2) \\ 4 \cdot (-1) & (-2) \cdot 3 \end{bmatrix}$$

When using `mmult`, you must make sure that when you multiply an $m \times n$ matrix with a $k \times h$ matrix, n must equal k . If not, the result will be a `#VALUE` message, which is Excel's way of telling you that your input is nonsense. You must also take care that the cells that you enter the answer into is an $m \times h$ matrix. If you enter the results into a larger matrix, you will get a matrix where a subset of the cells contains the correct answer and the rest of the entries will return `#N/A`, which is Excel's way of telling you that you are asking for output without providing input. The most important thing to note is that if you enter the result into a matrix that is too small, you will only get some of the elements, *and Excel will not provide you with any message indicating this.*

Example 1

Let the matrices **A** and **B** be given by:

$$\mathbf{A} = \begin{bmatrix} 1 & 0 & 3 \\ 2 & 1 & 5 \end{bmatrix}, \quad \mathbf{B} = \begin{bmatrix} 1 & 3 \\ 2 & 5 \\ 6 & 2 \end{bmatrix}$$

Find **AB** by using the `=mmult` command.

Solution:

First enter the matrices. Then note that the result must be a 2×2 matrix (why?). Enter the `=mmult` command into a 2×2 array of cells. The spreadsheet should look like Figure 8.

	A	B	C	D	E	F	G
1		A				B	
2		1	0	3		1	3
3		2	1	5		2	5
4						6	2
5							
6		C=AB					
7		<code>=mmult(B2:D3;F2:G4)</code>					
8							
9							

Figure 8

After pressing `<Ctrl> + <Shift> + <Enter>`, one can then verify that the result is $\begin{bmatrix} 19 & 9 \\ 34 & 21 \end{bmatrix}$.

A shortcoming of the `=mmult` command is that one can only multiply two matrices at the same time. However, one can nest the `=mmult` command. Let us say we want to find the matrix product **ABC** where **A**, **B**, and **C** are given in Figure 9.

	A	B	C	D	E	F	G	H	I
1		A			B			C	
2			1	4		1	-3		2
3			2	3		-1	2		-1
4									
5									
6		=mmult(B2:C3;mmult(E2:F3;H2:I3))							
7									
8									
9									

Figure 9. Multiplication of three matrices

Look at the array formula in the cells B6:C8. Here, the mmult command is used twice, so that the first mmult command has **A** as the first argument and the matrix product **BC** as the second. The answer is $\begin{bmatrix} -11 & 7 \\ -2 & 6 \end{bmatrix}$. You should verify that multiplying **AB** with **C** gives the same result. It is crucial that the order in which the matrices are entered into the formula is maintained. In this way, any matrix multiplication can be performed. Sometimes, we want to use a mix of absolute and relative cell references to do matrix multiplication.

Example 2

Compute **Ts**, **T(Ts)**, **T(T(Ts))**, ... , **T(T(T(T(Ts))))**. (See Example 15.3.3 in the textbook.)

First we compute **Ts**. Since **T** is 3×3 , and **s** is 3×1 , the result must be a 3×1 matrix. The matrices and the array formula are shown in Figure 10.

	A	B	C	D	E	F	G	H	I
14									
15		T				s	Ts		
16			0.85	0.1	0.1		0.2	=mmult(\$B\$16:\$D\$18;F\$16:F\$18)	
17			0.05	0.55	0.05		0.6		
18			0.1	0.35	0.85		0.2		
19									

Figure 10

Note how the reference to **T** uses an absolute cell reference, while the reference to **s** uses a mix of relative and absolute cell references. This is in order to simplify the calculations that follow. Pressing <Ctrl> + <Shift> + <Enter> gives us **Ts**. In order to calculate **T(Ts)**, the only thing we have to do is copy the contents of the cells F16:F18 and paste them into G16:G18. To find **T(T(Ts))**, we paste the formula into H16:H18 and so on. The result looks like Figure 11.

	E	F	G	H	I	J	K	L
14								
15		s	Ts	TTs	TTTs	TTTTs	TTTTTs	
16			0.2	0.25	0.2875	0.315625	0.336719	0.352539
17			0.6	0.35	0.225	0.1625	0.13125	0.115625
18			0.2	0.4	0.4875	0.521875	0.532031	0.531836
19								

Figure 11

You should verify that if we sum the elements in any column in Figure 11, they sum to 1. In particular is $0.352539063 + 0.115625 + 0.531835938 = 1$, so $\mathbf{T}(\mathbf{T}(\mathbf{T}(\mathbf{T}(\mathbf{T}s))))$ may be interpreted as the market share after 5 years.

Systems of equations in matrix form

In the textbook it is shown how a system like:

$$3x_1 + 4x_2 = 5$$

$$7x_1 - 2x_2 = 2$$

can be written more concisely as:

$$\begin{bmatrix} 3 & 4 \\ 7 & -2 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} = \begin{bmatrix} 5 \\ 2 \end{bmatrix}$$

Let us briefly illustrate how a system can be solved using the matrix formulation and the Solver. (Alternative methods will be considered later.) In Figure 12 we have illustrated how to enter such a system into a spreadsheet. The matrices \mathbf{A} , \mathbf{b} , and \mathbf{x} are entered as numbers. \mathbf{x} is our first guess on a solution to the system $\mathbf{Ax} = \mathbf{b}$.

	A	B	C	D	E	F	G	H	I
23									
24	A			x		Ax		b	
25	3	4		1		7		5	
26	7	-2		1		5		2	
27									
28									

Figure 12. Matrix representation of an equation system

\mathbf{Ax} is calculated with the command `=mmult(A25:B26;D25:D26)` assuming that $x_1 = 1$ and $x_2 = 2$. Obviously, $x_1 = 1$ and $x_2 = 2$ does not satisfy $\mathbf{Ax} = \mathbf{b}$. To solve the equations, start the Solver and ask it to maximise x_1 subject to the requirement that $\mathbf{Ax} = \mathbf{b}$. Since $x_1 = 1$ and $x_2 = 2$ are determined by the equations, the maximization will not impact the result. The Solver should look like Figure 13.




Figure 13. Solving linear systems with the Solver

Pressing Solve gives us the correct answers: $x_1 = 0.529411765$ and $x_2 = 0.852941176$.

Problems

1) Compute \mathbf{AB} and \mathbf{BA} , if possible, for the following:

a) $\mathbf{A} = \begin{bmatrix} 0 & -2 \\ 3 & 1 \end{bmatrix}$ and $\mathbf{B} = \begin{bmatrix} -1 & 4 \\ 1 & 5 \end{bmatrix}$

b) $\mathbf{A} = \begin{bmatrix} 8 & 3 & -2 \\ 1 & 0 & 4 \end{bmatrix}$ and $\mathbf{B} = \begin{bmatrix} 3 & 1 \\ -1 & 1 \\ 0 & 1 \end{bmatrix}$

2) In Example 2 above, repeat the process of copying cells until you have performed the calculation:

$$\underbrace{\mathbf{T}\mathbf{T}\mathbf{T}\cdots\mathbf{T}}_{50 \text{ times}}\mathbf{s}$$

What happens to the market shares as the number of years increases?

Rules for matrix multiplication

When doing calculations with a large number of matrices, it is helpful to define names for them. This makes the spreadsheet easier to read and reduces the likelihood of mistakes. To give a matrix a name, select the matrix, choose Insert and then Name in the menu. Then, choose Define and give the matrix a name. Any name is possible, but it is useful to use a single capital letter. When using the name in formulas, you must still take care to make sure that the dimensions you work with are the correct ones and that you use <Ctrl> + <Shift> + <Enter> whenever a matrix formula is entered. We illustrate the use of names in Example 3.

Example 3

Letting $\mathbf{A} = \begin{bmatrix} 1 & 0 \\ 2 & 1 \end{bmatrix}$, $\mathbf{B} = \begin{bmatrix} 1 & 3 \\ 2 & 5 \end{bmatrix}$ and $\mathbf{D} = \begin{bmatrix} 19 & 9 \\ 0 & 2 \end{bmatrix}$ (the last matrix is named \mathbf{D} since \mathbf{C} is a reserved symbol in Excel) we can verify the different rules by entering the following matrix formulas in Figure 14.

	A	B	C	D	E	F	G
7							
8		(AB)D				A(BD)	
9		=MMULT(MMULT(A;B);D)	=MMULT(MMULT(A;B);D)			=MMULT(A;MMULT(E=MMULT(A;MMULT(B;D)))	
10		=MMULT(MMULT(A;B);D)	=MMULT(MMULT(A;B);D)			=MMULT(A;MMULT(E=MMULT(A;MMULT(B;D)))	
11							
12		A(B+D)				AB+AD	
13		=MMULT(A;B+D)	=MMULT(A;B+D)			=MMULT(A;B)+MMUL=MMULT(A;B)+MMULT(A;D)	
14		=MMULT(A;B+D)	=MMULT(A;B+D)			=MMULT(A;B)+MMUL=MMULT(A;B)+MMULT(A;D)	
15							
16		(A+B)D				AD+BD	
17		=MMULT(A+B;D)	=MMULT(A+B;D)			=MMULT(A;D)+MMUL=MMULT(A;D)+MMULT(B;D)	
18		=MMULT(A+B;D)	=MMULT(A+B;D)			=MMULT(A;D)+MMUL=MMULT(A;D)+MMULT(B;D)	
19							

Figure 14. Matrix multiplication

You should study the formulas in Figure 14 carefully. Note how the associative, as well as the two distributive laws, are entered almost verbatim into the formulas. If the results are viewed as numbers rather than formulas, we see the same answers as in the main text.

Problems

- 1) Do Problem 15.4.1 of the textbook using Excel.
- 2) Do Problem 15.4.4 of the textbook using Excel.
- 3) Do Problem 15.4.5 of the textbook using Excel.
- 4) Do Problem 15.4.6(a) of the textbook using Excel.

The transpose

Excel has a function for transposing matrices. The syntax of this function is as follows: =TRANSPOSE(arg). You must take care to get the number of rows and columns correct. If you transpose an $m \times n$ matrix, you must select an $n \times m$ matrix in which to enter the results.

Example 4

Let $\mathbf{A} = \begin{bmatrix} -1 & 0 \\ 2 & 3 \\ 5 & -1 \end{bmatrix}$, $\mathbf{B} = \begin{bmatrix} 1 & -1 & 0 & 4 \\ 2 & 1 & 1 & 1 \end{bmatrix}$, Find \mathbf{A}' and \mathbf{B}' using Excel.

Solution:

Figure 15 shows the correct answer for \mathbf{A}' and how to enter the array formula for \mathbf{B}' .

	A	B	C	D	E	F	G
24		A			A'		
25		-1	0		-1	2	5
26		2	3		0	3	-1
27		5	-1				
28							
29							
30							
31	B					B'	
32	1	-1	0	4	=transpose(A32:D33)		
33	2	1	1	1			
34							
35							
36							

Figure 15. Calculating the transpose

Example 5

Let $\mathbf{x} = [1 \ -2 \ 3]'$. Note that \mathbf{x} is a column vector. Calculate $\mathbf{x}'\mathbf{x}$ and \mathbf{xx}' .

Solution:

Since \mathbf{x} is a 3×1 matrix, \mathbf{x}' is a 1×3 matrix. It follows that $\mathbf{x}'\mathbf{x}$ is a scalar, and \mathbf{xx}' is a 3×3 matrix. In order to find the solution, we combine the =mmult command and the =TRANSPOSE command. If \mathbf{x} is in the cell range B2:B4, \mathbf{xx}' can be calculated if the formula =MMULT(B2:B4;TRANSPOSE(B2:B4)) is entered into a 3×3 matrix. $\mathbf{x}'\mathbf{x}$ can be calculated by entering =MMULT(TRANSPOSE(B2:B4);B2:B4).

Example 6

Verify that \mathbf{AA}' and $\mathbf{A}'\mathbf{A}$ are symmetric for the matrix \mathbf{A} in Example 15.5.1.

Solution:

If \mathbf{A} is entered into the cell range A1:B3, then the following array formula =MMULT(TRANSPOSE(A1:B3);A1:B3) entered into a 2×2 matrix will calculate $\mathbf{A}'\mathbf{A}$, and =MMULT(A1:B3;TRANSPOSE(A1:B3)) entered into a 3×3 matrix will calculate \mathbf{AA}' . The results are shown in Figure 16. Note that in both cases the matrices are symmetric.

	A	B	C	D	E	F
4						
5	A'A			AA'		
6	30	1		1	-2	-5
7	1	10		-2	13	7
8				-5	7	26
9						
10						
11						

Figure 16

Problems

1) Do Problem 15.5.1 of the textbook using Excel.

2) Do Problem 15.5.2 of the textbook using Excel.

3) Let $\mathbf{A}_1 = \begin{bmatrix} 1 & 1 & 6 \\ -1 & 2 & 2 \\ 4 & 3 & -1 \end{bmatrix}$, $\mathbf{A}_2 = \begin{bmatrix} 2 & 1 & 6 \\ -3 & 0 & 2 \\ -4 & 3 & -5 \end{bmatrix}$ and $\mathbf{A}_3 = \begin{bmatrix} 8 & 1 & 1 \\ 0 & 0 & 5 \\ 3 & 3 & 0 \end{bmatrix}$

Verify that $(\mathbf{A}_1\mathbf{A}_2\mathbf{A}_3)' = (\mathbf{A}'_1\mathbf{A}'_2\mathbf{A}'_3)$.

4) Let \mathbf{P} be given by $\begin{bmatrix} \lambda & 0 & \lambda \\ \lambda & 0 & -\lambda \\ 0 & 1 & 0 \end{bmatrix}$. Enter $\mathbf{P}'\mathbf{P} = \mathbf{I}_3$ as a matrix equation and use the Solver to

find the values of λ that solve the equation.

Gaussian elimination

Let us examine the following problem:

$$\mathbf{Ax} = \begin{bmatrix} 2 & 3 & 5 \\ -2 & 3 & 1 \\ 6 & 2 & -1 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix} = \begin{bmatrix} 5 \\ 0 \\ \frac{1}{6} \end{bmatrix} = \mathbf{b} \quad (1)$$

We can perform the entire Gaussian elimination of the problem as in Figure 7.

	A	B	C	D	E	F	G	H
45								
46		2	3	5	5			
47		-2	3	1	0			
48		6	2	-1	0.166667			
49			~					
50		1	1.5	2.5	2.5	{=B46:E46/\$B\$460}		
51		0	6	6	5	{=B47:E47-(B47/\$B\$46)*\$B\$46:\$E\$46}		
52		0	-7	-16	-14.833333	{=B48:E48-(B48/\$B\$46)*\$B\$46:\$E\$46}		
53			~					
54		1	1.5	2.5	2.5	{=B50:E50}		
55		0	1	1	0.833333	{=B51:E51/\$C\$51}		
56		0	0	-9	-9	{=B52:E52-(C52/C51)*B51:E51}		
57			~					
58		1	1.5	2.5	2.5	{=B54:E54}		
59		0	1	1	0.833333	{=B55:E55}		
60		0	0	1	1	{=B56:E56/\$D\$56}		
61								

Figure 17. Gaussian elimination

The equation system is put in the same extended matrix form as in the text. The matrix in B46:E48 is the matrix $[A \ b]$. In the F column, the formulas are shown e.g. cell F50 shows the matrix formula entered into cells B50:E50. Entering the formulas in the F column as matrix formulas into the B, C, D and E columns will perform all the elementary row operations needed to create the upper triangular matrix in cells B58:E60. Note that the way in which the formulas are entered is general enough to reduce any system of equations to an upper triangular matrix as long as: a) a_{11} is not equal to 0; and b) a solution to the equations actually exists.

Problems

1) Solve the systems Problem 15.6.1 of the textbook using Gaussian elimination.

2) Consider the system:

$$\begin{aligned}x + y - z &= 1 \\x - y + 2z &= 2 \\x + 2y + az &= 3\end{aligned}$$

This system will have a unique solution except for a specific value of a . Set up the formulas that solve the system by Gaussian elimination. Use the Solver to find the value of a for which the system has no unique solution.

Vectors

We have already seen the matrix formulas that we can use to add, subtract and multiply matrices. The same formulas obviously work with vectors. Excel provides an additional function for the multiplication of vectors that will produce the inner product of two vectors regardless of whether the vectors are column vectors, row vectors or both. This function is `=SUMPRODUCT(arg1;arg2)`. This function does not have to be entered as an array formula. When entering the function, pressing `<Enter>` is enough. You can verify that the rules for the inner product holds for the `=SUMPRODUCT` function. Thus, if \mathbf{a} , \mathbf{b} and \mathbf{c} are n - vectors and t is a scalar, the rules for the inner product can be entered as follows:

$$\text{=SUMPRODUCT}(\mathbf{a};\mathbf{b}) \sim \text{=SUMPRODUCT}(\mathbf{b};\mathbf{a})$$

$$\text{=SUMPRODUCT}(\mathbf{a};\mathbf{b} + \mathbf{c}) \sim \text{=SUMPRODUCT}(\mathbf{a};\mathbf{b}) + \text{=SUMPRODUCT}(\mathbf{a};\mathbf{c})$$

$$\text{=SUMPRODUCT}(t*\mathbf{a};\mathbf{b}) \sim \text{=SUMPRODUCT}(\mathbf{a};t*\mathbf{b}) \sim t*\text{=SUMPRODUCT}(\mathbf{a};\mathbf{b})$$

Vectors have geometric interpretations, which are discussed in section 15.8 of the textbook. The length of a vector \mathbf{a} is calculated with the formula $\|\mathbf{a}\| = \sqrt{\mathbf{a} \cdot \mathbf{a}}$. This formula can be entered in Excel as `=SQRT(SUMPRODUCT(a;a))`. We can use this formula to examine the Cauchy-Schwarz inequality.

Example 7

For the two vectors $\mathbf{a} = (1, -2, 3)$ and $\mathbf{b} = (-3, 2, 5)$, check the Cauchy-Schwarz inequality.

Solution:

Figure 18 shows the formulas that are needed to verify the inequality. In cell B22 is the formula for $|\mathbf{a} \cdot \mathbf{b}|$. In cells B23 and B23 is the length of \mathbf{a} and \mathbf{b} . In cell B24, the product $\|\mathbf{a}\| \|\mathbf{b}\|$ is calculated. Switching back to viewing results, we find that $|\mathbf{a} \cdot \mathbf{b}| = 8$ $\|\mathbf{a}\| \|\mathbf{b}\| \approx 23.07$. Using similar formulas, we can calculate the cosine of the angle between two vectors. The formulas are shown in Figure 19.

	A	B	C	D	E	F
15		a		b		
16		1		-3		
17		-2		2		
18		3		5		
19						
20						
21						
22	$ \mathbf{a} \cdot \mathbf{b} $	=ABS(SUMPRODUCT(B16:B18;D16:D18))				
23						
24	$\ \mathbf{a}\ $	=SQRT(SUMPRODUCT(B16:B18;B16:B18))				
25	$\ \mathbf{b}\ $	=SQRT(SUMPRODUCT(D16:D18;D16:D18))				
26						
27	$\ \mathbf{a}\ \cdot \ \mathbf{b}\ $	=B24*B25				
28						
29						
30						

Figure 18. The Cauchy-Schwarz inequality

	A	B	C	D	E	F
15		a		b		
16		1		-3		
17		-2		2		
18		3		5		
19						
20	$\mathbf{a} \cdot \mathbf{b}$	=SUMPRODUCT(B16:B18;D16:D18)				
21	$\ \mathbf{a}\ $	=SQRT(SUMPRODUCT(B16:B18;B16:B18))				
22	$\ \mathbf{b}\ $	=SQRT(SUMPRODUCT(D16:D18;D16:D18))				
23						
24	$\mathbf{a} \cdot \mathbf{b}$					
25	$\frac{\mathbf{a} \cdot \mathbf{b}}{\ \mathbf{a}\ \cdot \ \mathbf{b}\ }$	=B20/(B21*B22)	θ		=ACOS(B25)	
26						

Figure 19. Geometric interpretations of vectors

In cells B20:22 are the formulas for $\mathbf{a} \cdot \mathbf{b}$, $\|\mathbf{a}\|$ and $\|\mathbf{b}\|$. Cell B25 has the cosine of the angle between then \mathbf{a} and \mathbf{b} (alternatively the correlation coefficient). In cell E25, the acos function

is used to calculate the actual angle.² The results will be $\cos(\theta) = 0.346843988$ and $\theta = 1.216592$. Note that θ is measured in radians.

Problems

- 1) Solve Problem 15.7.1 of the textbook.
- 2) Solve Problem 15.7.2 of the textbook.
- 3) Use the Solver to find x , y and z in Problem 15.7.2 of the textbook.
- 4) Solve Problem 15.7.7 of the textbook.
- 5) Solve Problem 15.7.8 of the textbook using the Solver.
- 6) Solve Problem 15.8.1 of the textbook.

Determinants and the inverse of a matrix

The determinant

When computing determinants, you will quickly see the use of computers. The determinant of a 5×5 matrix consists of 120 terms. Doing this calculation by hand requires patience, pedantry and lots of time. Finding the determinant of a 5×5 matrix using Excel takes seconds. In order to find a determinant using Excel, you must use the `=mdeterm(arg)` function. The argument should be an $n \times n$ range. The `=mdeterm()` function is not an array function, so you can enter the function as a standard Excel function.

Example 8

Find the determinant of the matrix $\mathbf{A} = \begin{bmatrix} 2 & -4 \\ 3 & 6 \end{bmatrix}$.

Solution:

Enter the matrix into the range A1:B2. Then, enter the formula `=mdeterm(A1:B2)` into a cell. The result should be 24.

The inverse

You can compute inverse matrices using Excel. You must use the function `=minverse(arg)`. This is an array function, so you must enter the function as described above.

² Note that the angle is calculated in radians rather degrees.

Example 9

Find the inverse of the matrix $\mathbf{A} = \begin{bmatrix} 2 & -4 \\ 3 & 6 \end{bmatrix}$.

Solution:

Enter the matrix into the range A1:B2. Select a range that has two rows and two columns. Then, enter the formula =minverse(A1:B2) into the range. Press <Ctrl> + <Shift> + <Enter>. The answer should look like Figure 20. Note that the formula for the inverse is visible in the formula bar.

	A	B	C	D	E	F
14						
15		A			A⁻¹	
16		2	-4		0.25	0.166667
17		3	6		-0.125	0.083333
18						

Figure 20. Computing the inverse

One of the uses of the inverse is to solve systems of linear equations, since $\mathbf{Ax} = \mathbf{b}$ implies $\mathbf{x} = \mathbf{A}^{-1}\mathbf{b}$ if the inverse exists. You can use this formula to solve linear equations using Excel, as shown in Example 10.

Example 10

Let $\mathbf{A} = \begin{bmatrix} 2 & 1 \\ 2 & 2 \end{bmatrix}$ and $\mathbf{b} = \begin{bmatrix} 3 \\ 4 \end{bmatrix}$. Let $\mathbf{Ax} = \mathbf{b}$. Find \mathbf{x} .

Solution:

Enter \mathbf{A} and \mathbf{b} into separate cell ranges. In order to calculate $\mathbf{A}^{-1}\mathbf{b}$, you must use the =mmult command to multiply \mathbf{A}^{-1} and \mathbf{b} . If \mathbf{A} is entered into A1:B2, and \mathbf{b} is entered into D1:D2, then enter the formula =mmult(minverse(A1:A2);D1:D2) into a range with two rows and one column. The answer should be:

$$\mathbf{x} = \mathbf{A}^{-1}\mathbf{b} = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

When solving systems of linear equations using Excel, you can choose between a number of methods. The textbook presents three methods. Gaussian elimination, Cramer's method, and solving equations by matrix inversion. In Excel, solving equations by matrix inversion is by far the simplest method. Cramer's rule is so cumbersome that it is not recommended. However, sometimes Excel produces errors. Excel rounds numbers, and will occasionally compute \mathbf{A}^{-1} even if a matrix has a determinant equal to zero. Also, if a determinant is close to zero, Excel may internally round numbers so that the inverse is incorrect. If this happens, your solution will be wrong. You should always examine the determinant. If the determinant is

close to zero, you should try to verify the solution using other methods. For instance, you can always try to solve the system using the Solver.

Problems

- 1) Solve Problem 16.5.1 of the textbook.
- 2) Solve Problem 16.5.2(c) of the textbook.
- 3) Consider the matrix:

$$\mathbf{A} = \begin{bmatrix} 1 & 1 & 2 \\ -3 & \lambda & -\frac{4}{3} \\ 4 & 1 & 10 \end{bmatrix}$$

Use the =mdeterm function and the Solver to find the value of λ that makes $\det(\mathbf{A}) = 0$.

- 4) Solve Problems 16.6.1 and 16.6.2 of the textbook.
- 5) Solve Problems 16.6.6(a) and (b) of the textbook.